F07PEF (SSPTRS/DSPTRS) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07PEF (SSPTRS/DSPTRS) solves a real symmetric indefinite system of linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F07PDF (SSPTRF/DSPTRF), using packed storage.

2 Specification

```
SUBROUTINE F07PEF(UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
ENTRY      ssptrs(UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
INTEGER            N, NRHS, IPIV(*), LDB, INFO
real               AP(*), B(LDB,*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To solve a real symmetric indefinite system of linear equations $AX = B$, this routine must be preceded by a call to F07PDF (SSPTRF/DSPTRF) which computes the Bunch–Kaufman factorization of A using packed storage.

If UPLO = 'U', $A = PUDU^TP^T$, where P is a permutation matrix, U is an upper triangular matrix and D is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution X is computed by solving $PUDY = B$ and then $U^TP^TX = Y$.

If UPLO = 'L', $A = PLDL^TP^T$, where L is a lower triangular matrix; the solution X is computed by solving $PLDY = B$ and then $L^TP^TX = Y$.

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates how A has been factorized as follows:

if UPLO = 'U', then $A = PUDU^TP^T$, where U is upper triangular;
if UPLO = 'L', then $A = PLDL^TP^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: NRHS — INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: NRHS ≥ 0.

4: AP(∗) — ***real*** array *Input*

Note: the dimension of the array AP must be at least max(1,N∗(N+1)/2).

On entry: details of the factorization of A stored in packed form, as returned by F07PDF (SSPTRF/DSPTRF).

5: IPIV(∗) — INTEGER array *Input*

Note: the dimension of the array IPIV must be at least max(1,N).

On entry: details of the interchanges and the block structure of D, as returned by F07PDF (SSPTRF/DSPTRF).

6: B(LDB,∗) — ***real*** array *Input/Output*

Note: the second dimension of the array B must be at least max(1,NRHS).

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

7: LDB — INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07PEF (SSPTRS/DSPTRS) is called.

Constraint: LDB ≥ max(1,N).

8: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$\begin{array}{ll} |E| \leq c(n)\epsilon P|U||D||U^T|P^T & \text{if UPLO = 'U',} \\ |E| \leq c(n)\epsilon P|L||D||L^T|P^T & \text{if UPLO = 'L',} \end{array}$$

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\text{cond}(A, x)\epsilon$$

where $\text{cond}(A, x) = \| \, |A^{-1}||A||x| \, \|_\infty / \|x\|_\infty \leq \text{cond}(A) = \| \, |A^{-1}||A| \, \|_\infty \leq \kappa_\infty(A)$. Note that $\text{cond}(A, x)$ can be much smaller than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07PHF (SSPRFS/DSPRFS), and an estimate for $\kappa_\infty(A)$ $(= \kappa_1(A))$ can be obtained by calling F07PGF (SSPCON/DSPCON).

8 Further Comments

The total number of floating-point operations is approximately $2n^2r$.

This routine may be followed by a call to F07PHF (SSPRFS/DSPRFS) to refine the solution and return an error estimate.

The complex analogues of this routine are F07PSF (CHPTRS/ZHPTRS) for Hermitian matrices and F07QSF (CSPTRS/ZSPTRS) for symmetric matrices.

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix} \quad \textit{and} \quad B = \begin{pmatrix} -9.50 & 27.85 \\ -8.38 & 9.90 \\ -6.07 & 19.25 \\ -0.96 & 3.93 \end{pmatrix}.$$

Here A is symmetric indefinite, stored in packed form, and must first be factorized by F07PDF (SSPTRF/DSPTRF).

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07PEF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NRHMAX, LDB
      PARAMETER        (NMAX=8,NRHMAX=NMAX,LDB=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N, NRHS
      CHARACTER        UPLO
*     .. Local Arrays ..
      real             AP(NMAX*(NMAX+1)/2), B(LDB,NRHMAX)
      INTEGER          IPIV(NMAX)
*     .. External Subroutines ..
      EXTERNAL         ssptrf, ssptrs, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07PEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
         END IF
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Factorize A
*
         CALL ssptrf(UPLO,N,AP,IPIV,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
```

```
*           Compute solution
*
            CALL ssptrs(UPLO,N,NRHS,AP,IPIV,B,LDB,INFO)
*
*           Print solution
*
            IFAIL = 0
*
            CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution(s)',IFAIL)
*
         ELSE
            WRITE (NOUT,*) 'The factor D is singular'
         END IF
      END IF
      STOP
*
      END
```

9.2 Program Data

```
F07PEF Example Program Data
  4  2                        :Values of N and NRHS
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
 -9.50  27.85
 -8.38   9.90
 -6.07  19.25
 -0.96   3.93                 :End of matrix B
```

9.3 Program Results

```
 F07PEF Example Program Results

 Solution(s)
             1          2
 1     -4.0000     1.0000
 2     -1.0000     4.0000
 3      2.0000     3.0000
 4      5.0000     2.0000
```